

Math 2250
Maple Project 2 Part A: Linear Algebra
February 2005

Due date: See the internet due dates. Maple lab 2 has parts A (problems 2.1, 2.2, 2.3) and B (problems 2.4, 2.5, 2.6), issued in two different documents. This document is part A.

References: Code in `maple` appears in `2250mapleL2a-S2005.txt` at URL <http://www.math.utah.edu/~gustafso/>. This document: `2250mapleL2a-S2005.pdf`.

Problem 2.1. (Matrix Algebra)

Define $A = \begin{pmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \\ 7 & 8 & 9 \end{pmatrix}$, $B = \begin{pmatrix} 2 & 1 & 0 \\ 1 & 2 & 1 \\ 0 & 1 & 2 \end{pmatrix}$, $\mathbf{v} = \begin{pmatrix} 1 \\ 2 \\ 3 \end{pmatrix}$ and $\mathbf{w} = \begin{pmatrix} -1 \\ 4 \\ 1 \end{pmatrix}$. Create a worksheet in `maple` which states this problem in text, then defines the four objects. The worksheet should contain text, `maple` code and displays. Continue with this worksheet to answer (1)–(7) below. Submit problem 2.1 as a worksheet print on 8.5 by 11 inch paper. See Example 1 for `maple` commands.

- (1) Compute AB and BA . Are they the same?
- (2) Compute $A + B$ and $B + A$. Are they the same?
- (3) Let $C = A + B$. Compare C^2 to $A^2 + 2AB + B^2$. Explain why they are different.
- (4) Compute the transpose of AB and compare it to the product of the transpose of A with the transpose of B , multiplied in the correct order so that you expect equality.
- (5) Solve for $\mathbf{X}$ in $B\mathbf{X} = \mathbf{v}$ by three different methods.
- (6) Solve $A\mathbf{Y} = \mathbf{v}$ for $\mathbf{Y}$. Do an answer check.
- (7) Solve $A\mathbf{Z} = \mathbf{w}$. Explain your answer.

Problem 2.2. (Row space)

Let $A = \begin{pmatrix} 1 & 1 & 1 & 2 & 6 \\ 2 & 3 & -2 & 1 & -3 \\ 0 & 1 & -4 & -3 & -15 \\ 1 & 2 & -3 & -1 & -9 \end{pmatrix}$. Find three different bases for the row space of A , using the following methods.

1. The method of Example 2, below.
2. The `maple` command `rowSPACE(A)`.
3. The `rref`-method: select rows from `rref(A)`.

Verify that all three bases are equivalent.

Problem 2.3. (Matrix Equations)

Let $A = \begin{pmatrix} 8 & 10 & 3 \\ -3 & -5 & -3 \\ -4 & -4 & 1 \end{pmatrix}$, $T = \begin{pmatrix} 1 & 0 & 0 \\ 0 & -2 & 0 \\ 0 & 0 & 5 \end{pmatrix}$. Let P denote a 3×3 matrix. Assume the following result (proved below):

Lemma 1. The equality $AP = PT$ holds if and only if the columns $\mathbf{v}_1$, $\mathbf{v}_2$, $\mathbf{v}_3$ of P satisfy $A\mathbf{v}_1 = \mathbf{v}_1$, $A\mathbf{v}_2 = -2\mathbf{v}_2$, $A\mathbf{v}_3 = 5\mathbf{v}_3$.

- (a) Determine three specific columns for P such that $\det(P) \neq 0$ and $AP = PT$. Infinitely many answers are possible. See Example 4 for the `maple` method that determines a column of P . (b) After reporting the three columns, check the answer by computing $AP - PT$ (it should be zero) and $\det(P)$ (it should be nonzero).

Example 1. Let $A = \begin{pmatrix} 1 & 2 & 3 \\ 2 & -1 & 1 \\ 3 & 0 & -1 \end{pmatrix}$ and $\mathbf{b} = \begin{pmatrix} 9 \\ 8 \\ 3 \end{pmatrix}$. Define and display matrix A and vector $\mathbf{b}$ in `maple`. Then compute

- (1) The inverse of A .
- (2) The augmented matrix $C = \text{aug}(A, \mathbf{b})$.
- (3) The reduced row echelon form $R = \text{rref}(C)$.
- (4) The column $\mathbf{X}$ of R which solves $A\mathbf{X} = \mathbf{b}$.
- (5) The matrix A^3 .
- (6) The transpose of A .
- (7) The matrix $AC - 3A^2$.
- (8) The solution $\mathbf{X}$ of $A\mathbf{X} = \mathbf{b}$ by two methods different than (4).

Solution: To get help, enter `?linalg` into a worksheet, then select commands that match ones below.

```
with(linalg):
A:=matrix([[1,2,3],[2,-1,1],[3,0,-1]]);
b:=vector([9,8,3]);
print("(1)"); inverse(A);
print("(2)"); C:=augment(A,b);
print("(3)"); R:=rref(C);
print("(4)"); X:=col(R,4);
print("(5)"); evalm(A^3);
print("(6)"); transpose(A);
print("(7)"); evalm(A&*C-3*(A^2));
print("(8)"); X:=linsolve(A,b); X:=evalm(inverse(A) &* b);
```

Example 2. Let $A = \begin{pmatrix} 1 & 1 & 1 & 2 & 6 \\ 2 & 3 & -2 & 1 & -3 \\ 3 & 5 & -5 & 1 & -8 \\ 4 & 3 & 8 & 2 & 3 \end{pmatrix}$.

- (1) Find a basis for the column space of A .
- (2) Find a basis for the row space of A .
- (3) Find a basis for the nullspace of A .
- (4) Find $\text{rank}(A)$ and $\text{nullity}(A)$.
- (5) Find the dimensions of the nullspace, row space and column space of A .

Solution: The theory applied: *The columns of B corresponding to the leading ones in $\text{rref}(B)$ are independent and form a basis for the column space of B .* Results for the row space can be obtained by applying the above theory to the transpose of the matrix.

The `maple` code which applies is

```
with(linalg):
A:=matrix([[ 1, 1, 1, 2, 6],
            [ 2, 3,-2, 1,-3],
            [ 3, 5,-5, 1,-8],
            [ 4, 3, 8, 2, 3]]);
print("(1)"); C:=rref(A); # leading ones on columns 1,2,4
BASIScolumnspace=col(A,1),col(A,2),col(A,4);
print("(2)"); F:=rref(transpose(A)); # leading ones on columns 1,2,3
```

```

BASISrowSpace=row(A,1),row(A,2),row(A,3);
print("(3)"); nullspace(A); linsolve(A,vector([0,0,0,0]));
print("(4)"); RANK=rank(A); NULLITY=coldim(A)-rank(A);
print("(5)"); DIMnullspace=coldim(A)-rank(A); DIMrowSpace=rank(A);
DIMcolumnSpace=rank(A);

```

Example 3. Let $A = \begin{pmatrix} 1 & 1 & 1 & 2 & 6 \\ 2 & 3 & -2 & 1 & -3 \\ 3 & 5 & -5 & 1 & -8 \\ 4 & 3 & 8 & 2 & 3 \end{pmatrix}$. Verify that the following column space bases of A are equivalent.

$$\mathbf{v}_1 = \begin{pmatrix} 1 \\ 2 \\ 3 \\ 4 \end{pmatrix}, \quad \mathbf{v}_2 = \begin{pmatrix} 1 \\ 3 \\ 5 \\ 3 \end{pmatrix}, \quad \mathbf{v}_3 = \begin{pmatrix} 2 \\ 1 \\ 1 \\ 2 \end{pmatrix},$$

$$\mathbf{w}_1 = \begin{pmatrix} 1 \\ 0 \\ 0 \\ -3 \end{pmatrix}, \quad \mathbf{w}_2 = \begin{pmatrix} 0 \\ 1 \\ 0 \\ 17 \end{pmatrix}, \quad \mathbf{w}_3 = \begin{pmatrix} 0 \\ 0 \\ 1 \\ -9 \end{pmatrix}.$$

Solution: The maple code which applies is

```

with(linalg):
A:=matrix([[ 1, 1, 1, 2, 6],
            [ 2, 3,-2, 1,-3],
            [ 3, 5,-5, 1,-8],
            [ 4, 3, 8, 2, 3]]);
v1:=vector([1, 2, 3, 4]); v2:=vector([1, 3, 5, 3]); v3:=vector([2, 1, 1, 2]);
w1:=vector([1, 0, 0, -3]); w2:=vector([0, 1, 0, 17]); w3:=vector([0, 0, 1, -9]);
F:=augment(v1,v2,v3);
G:=augment(w1,w2,w3);
rank(A); rank(F); rank(G);
rank(augment(A,v1))-rank(A);
rank(augment(A,v2))-rank(A);
rank(augment(A,v3))-rank(A);
rank(augment(A,w1))-rank(A);
rank(augment(A,w2))-rank(A);
rank(augment(A,w3))-rank(A);

```

The theory says that two bases of a subspace S of $\mathcal{R}^4$ are equivalent. We justify that the proposed sets are independent and have the correct size ($\text{rank}(F) = \text{rank}(G) = \text{rank}(A)$), and that all six vectors are in the column space of A ($\text{rank}(\text{aug}(A, \mathbf{v})) = \text{rank}(A)$ if and only if $A\mathbf{X} = \mathbf{v}$ has a solution $\mathbf{X}$).

Example 4. Let $A = \begin{pmatrix} 1 & 2 & 3 \\ 2 & -1 & 1 \\ 3 & 0 & 0 \end{pmatrix}$. Solve the equation $A\mathbf{x} = -3\mathbf{x}$ for $\mathbf{x}$.

The maple details appear below. The idea is to write the problem as a homogeneous problem $(A + 3I)\mathbf{x} = \mathbf{0}$, which always has a solution.

```

with(linalg):
A:=matrix([[1,2,3],[2,-1,1],[3,0,0]]);
linsolve(evalm(A+3*diag(1,1,1)),vector([0,0,0]));
# ans: t_1*vector([-2,1,2])

```

Proof of Lemma 1. Define $r_1 = 1$, $r_2 = -2$, $r_3 = 5$. Assume $AP = PT$, $P = \text{aug}(\mathbf{v}_1, \mathbf{v}_2, \mathbf{v}_3)$ and $T = \text{diag}(r_1, r_2, r_3)$. The definition of matrix multiplication implies that $AP = \text{aug}(A\mathbf{v}_1, A\mathbf{v}_2, A\mathbf{v}_3)$ and $PT = \text{aug}(r_1\mathbf{v}_1, r_2\mathbf{v}_2, r_3\mathbf{v}_3)$. Then $AP = PT$ holds if and only if the columns of the two matrices match, which is equivalent to the three equations $A\mathbf{v}_1 = r_1\mathbf{v}_1$, $A\mathbf{v}_2 = r_2\mathbf{v}_2$, $A\mathbf{v}_3 = r_3\mathbf{v}_3$. The proof is complete.

End of Maple Lab 2 Part A.